

PRESEK

List za mlade matematike, fizike, astronome in računalnikarje

ISSN 0351-6652

Letnik 12 (1984/1985)

Številka 4

Strani 163-166

Tomaž Pisanski:

HI KVADRAT Z RAČUNALNIKOM

Ključne besede: računalništvo.

Elektronska verzija: <http://www.presek.si/12/731-Pisanski.pdf>

© 1985 Društvo matematikov, fizikov in astronomov Slovenije

© 2009 DMFA - založništvo

Vse pravice pridržane. Razmnoževanje ali reproduciranje celote ali posameznih delov brez poprejšnjega dovoljenja založnika ni dovoljeno.

HI KVADRAT Z RAČUNALNIKOM

V prispevku z naslovom TEST HI KVADRAT smo spoznali statistično metodo, ki nam omogoča primerjati dejansko porazdelitev s teoretično. Med drugim lahko z njo ugotavljamo, ali je kocka, ki jo uporabljamo pri igri človek ne jezi se, poštena ali ne. Omenjeni članek je objavljen na str. 204.

Spomnimo se testa hi kvadrat. Denimo, da ima poskus n izidov. Naj bodo O_1, O_2, \dots, O_n dejanske absolutne frekvence s katerimi se pojavljajo posamezni izidi pri ponavljanju poskusa. Naj bodo E_1, E_2, \dots, E_n teoretične absolutne frekvence. Izraz

$$\chi^2(n-1) = (E_1 - O_1)^2/E_1 + (E_2 - O_2)^2/E_2 + \dots + (E_n - O_n)^2/E_n$$

imenujemo hi kvadrat z $(n - 1)$ prostostnimi stopnjami. V prejšnjem prispevku smo spoznali, kako lahko uporabimo izračunano vrednost.

V tem prispevku pa si oglejmo program v basicu, ki med drugim računa vrednosti hi kvadrat. Medtem ko program preverja, ali je vgrajeni generator slučajnih števil dober, posebni podprogram računa vrednost hi kvadrat. Bralci, ki želijo uporabiti program, ga morajo prilagoditi možnostim, ki jih ima njihov računalnik. Za lep izpis morajo upoštevati na primer število znakov v vrstici.

Oglejmo si nekaj značilnosti tega programa.

Vrstice 100 - 230: Glava programa. Vsebuje osnovne podatke o programu, pomemben del programske dokumentacije. Namesto komentarjev REM uporabljamo kar izpis PRINT. Rezerviramo prostor za največ 100 izidov.

Vrstice 240 - 310: Program sestoji iz treh klicev podprogramov: priprava podatkov, računanje in izpis rezultatov. Program simulira metanje kocke. Basic, ki ga uporabljamo, ima

```

100 REM      HI KVADRAT, programiral T.P., dne 24.1.1985
160 PRINT
170 PRINT " Računalnik meče 'kocko' z N ( <= 100 ) ploskvami"
180 PRINT
190 PRINT " O(1), ..., O(N) ... Dejanske frekvence"
200 PRINT " E(1), ..., E(N) ... Teoretične frekvence"
210 PRINT " M          ... Število poskusov"
220 PRINT
230   DIM O(100), E(100)
240 PRINT " Priprava podatkov "
250   GOSUB 1000
260 PRINT " Računamo hi kvadrat "
270   GOSUB 2000
280 PRINT " Izpis rezultatov "
290   GOSUB 3000
300 PRINT " Konec "
310   STOP
1000 REM Priprava podatkov .....
1010   INPUT " Slučajno število ";X
1020   RANDOMIZE(X)
1030   INPUT " Število izidov (6 za navadno kocko) ";N
1040   IF N < 2 OR N > 100 THEN GOTO 1030
1050   INPUT " Število poskusov ";M
1060   IF M < 5*N THEN GOTO 1050
1070   FOR I = 1 TO N
1080     LET O(I) = 0
1090     LET E(I) = M/N
1100   NEXT I
1110   FOR I = 1 TO M
1120     LET J = INT(RND*N) + 1
1130     LET O(J) = O(J) + 1
1140   NEXT I
1150   RETURN
2000 REM računanje hi kvadrat .....
2010   LET H2 = 0
2020   FOR I=1 TO N
2030     LET H2 = H2 + (O(I) - E(I))^2/E(I)
2040   NEXT I
2050   RETURN
3000 REM Izpis rezultatov .....
3010   PRINT
3020   PRINT "Izid          Dejanska          Teoretična "
3030   PRINT "          frekvenca          frekvenca "
3040   PRINT "*****"
3050   FOR I = 1 TO N
3060     PRINT I,O(I),E(I)
3070   NEXT I
3090   PRINT "*****"
3100   PRINT
3110   PRINT "Število izidov          = ",N
3120   PRINT "Število poskusov      = ",M
3130   PRINT "Število prostostnih stopenj = ",N-1
3140   PRINT "Vrednost hi kvadrat    = ",H2
3150   RETURN

```

vgrajeno funkcijo RND, ki vrne slučajno število med 0 in 1. V resnici zaporedje klicev funkcije RND ne vrača slučajnega zaporedja števil ampak le zaporedje števil, ki se vede, kot da bi bilo slučajno.

Vrstice 1000 - 1150: Podprogram za branje in pripravo podatkov. Določi podatke N, M, O in E. Če bi želeli z računalnikom računati hi kvadrat za primere iz prispevka TEST HI KVADRAT, bi morali ta podprogram spremeniti, tako da bi bral podatke O in E.

Vrstice 2000 - 2050: Računanje hi kvadrat. Iz N, M, O in E naračunamo hi kvadrat H2.

Vrstice 3000 - 3150: Podprogram za izpis rezultatov. Iz priloženih primerov vidimo, da še ni v najlepši obliki. Tako smo izpisali kar pet decimalk števila hi kvadrat, čeprav bi zadoščali že dve decimalki. Seveda pa bi lepšanje izpisa podaljšalo program, zato take izboljšave prepuščamo bralcem.

Računalnik meče 'kocko' z N (≤ 100) ploskvami

O(1), ..., O(N) ... Dejanske frekvence
 E(1), ..., E(N) ... Teoretične frekvence
 M ... Število poskusov

Priprava podatkov
 Slučajno število ? 1984
 Število izidov (6 za navadno kocko) ? 6
 Število poskusov ? 600
 Računamo hi kvadrat
 Izpis rezultatov

Izid	Dejanska frekvenca	Teoretična frekvenca
1	111	100
2	99	100
3	107	100
4	85	100
5	103	100
6	95	100

Število izidov	=	6
Število poskusov	=	600
Število prostostnih stopenj	=	5
Vrednost hi kvadrat	=	4.30002
Konec		

Računalnik meče 'kocko' z N (≤ 100) ploskvami

O(1), ..., O(N) ... Dejanske frekvence
E(1), ..., E(N) ... Teoretične frekvence
M ... Število poskusov

Priprava podatkov

Slučajno število ? 112

Število izidov (ó za navadno kocko) ? 10

Število poskusov ? 1000

Računamo hi kvadrat

Izpis rezultatov

Izid	Dejanska frekvenca	Teoretična frekvenca
1	105	100
2	94	100
3	93	100
4	101	100
5	98	100
6	104	100
7	95	100
8	103	100
9	101	100
10	106	100

```
Število izidov           =      10
Število poskusov        =     1000
Število prostostnih stopenj =      9
Urednost hi kvadrat     =     2.02001
Konec
```

Bralca vabimo, da pogleda v razpredelnici iz prejšnjega prispevka, ali je generator slučajnih števil prestal preprost statistični test. Če ga ni, mu nikakor ne smemo zaupati in moramo poiskati boljšega. Če pa je generator test uspešno prestal, pa to še ne pomeni, da je dober, saj bi se prav lahko "spotaknil" ob kakšnem drugačnem testu. O generatorjih slučajnih števil bomo spregovorili kdaj drugič.

Tomaž Pisanski